Computergestuetzte Mathematik zur Analysis

Lektion 12 (21. Januar)

```
> restart:
```

Extrema unter Nebenbedingungen

```
> with(plots):
> with(algcurves):
> with(VectorCalculus):
> BasisFormat(false):
> g := x^4 + y^4 - 1;
```

$$g := x^4 + y^4 - 1 \quad (1.1)$$

```
> NB := plot_real_curve(g, x, y):
> display(NB, scaling = constrained, thickness = 2);
```




```
> f := x + y;
```

$$f := x + y \tag{1.2}$$

```
> Nf := seq(solve(f = k, y),k=-1..1);
```

$$Nf := -1 - x, -x, 1 - x \tag{1.3}$$

```
> pl_NF:= plot([Nf], x = -1 .. 1, thickness = 2): # 1. Frage
> display({NB, pl_NF}, scaling = constrained);
```




```
> grad_g := Gradient(g, [x, y]);
```

$$grad\_g := \begin{bmatrix} 4\,x^3 \\ 4\,y^3 \end{bmatrix} \quad \mathbf{(1.4)}$$

```
> grad_f := Gradient(f, [x, y]);
```

$$grad\_f := \begin{bmatrix} 1 \\ 1 \end{bmatrix} \quad \mathbf{(1.5)}$$

```
> GLF := grad_f - lambda*grad_g; # Gradient Lagrangefunktion
```

$$GLF := \begin{bmatrix} -4\,\lambda\,x^3+1 \\ -4\,\lambda\,y^3+1 \end{bmatrix} \quad \mathbf{(1.6)}$$

```
> M:={solve({GLF[1] = 0, GLF[2] = 0, g = 0}),{x,y,lambda}};
```

$$M := \Big\{ \{\lambda, x, y\}, \Big\{\lambda = -\frac{1}{4}\,(RootOf(\_Z^2-\_Z+1)-1)\,RootOf(\_Z^4 \quad \mathbf{(1.7)}$$

$$-RootOf(\_Z^2-\_Z+1)),\, x = RootOf(\_Z^4-RootOf(\_Z^2-\_Z+1)),\, y =$$

$-RootOf(\_Z^4 - RootOf(\_Z^2 - \_Z + 1))\ RootOf(\_Z^2 - \_Z + 1)\Big\}, \Big\{\lambda = \frac{1}{2}\ RootOf(2\ \_Z^4 - 1), x = RootOf(2\ \_Z^4 - 1), y = RootOf(2\ \_Z^4 - 1)\Big\}\Big\}$

```
> M1 := M[1];
```

$$M1 := \{\lambda, x, y\} \tag{1.8}$$

```
> M2 := M[2];
```

$$M2 := \Big\{\lambda = -\frac{1}{4}\ (RootOf(\_Z^2 - \_Z + 1) - 1)\ RootOf(\_Z^4 - RootOf(\_Z^2 - \_Z + 1)), x = RootOf(\_Z^4 - RootOf(\_Z^2 - \_Z + 1)), y = -RootOf(\_Z^4 - RootOf(\_Z^2 - \_Z + 1))\ RootOf(\_Z^2 - \_Z + 1)\Big\} \tag{1.9}$$

```
> subs(M2, x);
```

$$RootOf(\_Z^4 - RootOf(\_Z^2 - \_Z + 1)) \tag{1.10}$$

```
> allvalues((1.10));
```

$$\frac{1}{2}\sqrt{2\,I + 2\sqrt{3}}, -\frac{1}{2}\sqrt{-2\,I - 2\sqrt{3}}, -\frac{1}{2}\sqrt{2\,I + 2\sqrt{3}}, \frac{1}{2}\sqrt{-2\,I - 2\sqrt{3}}, \frac{1}{2}\sqrt{-2\,I + 2\sqrt{3}}, \frac{1}{2}\sqrt{2\,I - 2\sqrt{3}}, -\frac{1}{2}\sqrt{-2\,I + 2\sqrt{3}}, -\frac{1}{2}\sqrt{2\,I - 2\sqrt{3}} \tag{1.11}$$

```
> simplify(evalc([(1.11)]));
```

$$\Big[\frac{1}{4}\sqrt{2}\sqrt{3} + \frac{1}{4}\sqrt{2} + \frac{1}{4}I\sqrt{2}\sqrt{3} - \frac{1}{4}I\sqrt{2}, -\frac{1}{4}\sqrt{2}\sqrt{3} + \frac{1}{4}\sqrt{2} + \frac{1}{4}I\sqrt{2}\sqrt{3} + \frac{1}{4}I\sqrt{2}, -\frac{1}{4}\sqrt{2}\sqrt{3} - \frac{1}{4}\sqrt{2} - \frac{1}{4}I\sqrt{2}\sqrt{3} + \frac{1}{4}I\sqrt{2}, \frac{1}{4}\sqrt{2}\sqrt{3} - \frac{1}{4}\sqrt{2} - \frac{1}{4}I\sqrt{2}\sqrt{3} - \frac{1}{4}I\sqrt{2}, \frac{1}{4}\sqrt{2}\sqrt{3} + \frac{1}{4}\sqrt{2} - \frac{1}{4}I\sqrt{2}\sqrt{3} + \frac{1}{4}I\sqrt{2}, \frac{1}{4}\sqrt{2}\sqrt{3} - \frac{1}{4}\sqrt{2} + \frac{1}{4}I\sqrt{2}\sqrt{3} + \frac{1}{4}I\sqrt{2}, -\frac{1}{4}\sqrt{2}\sqrt{3} - \frac{1}{4}\sqrt{2} + \frac{1}{4}I\sqrt{2}\sqrt{3} - \frac{1}{4}I\sqrt{2}, -\frac{1}{4}\sqrt{2}\sqrt{3} + \frac{1}{4}\sqrt{2} - \frac{1}{4}I\sqrt{2}\sqrt{3} - \frac{1}{4}I\sqrt{2}\Big] \tag{1.12}$$

Alle diese Loesungen haben einen nicht-verschwindenden Imaginaerteil. Zur Kontrolle

```
> evalf(%, 2);
```

$$[0.95 + 0.25\,I, -0.25 + 0.95\,I, -0.95 - 0.25\,I, 0.25 - 0.95\,I, 0.95 - 0.25\,I, 0.25 + 0.95\,I, -0.95 + 0.25\,I, -0.25 - 0.95\,I] \tag{1.13}$$

```
> M3:= M[3];
```

(1.14)

$$M3 := \left\{\lambda = \frac{1}{2}\, RootOf(2\_Z^4 - 1),\ x = RootOf(2\_Z^4 - 1),\ y = RootOf(2\_Z^4 - 1)\right\} \quad (1.14)$$

```
> allvalues(subs(M3,x));
```

$$\frac{1}{2}\,2^{3/4},\ \frac{1}{2}\,I\,2^{3/4},\ -\frac{1}{2}\,2^{3/4},\ -\frac{1}{2}\,I\,2^{3/4} \quad (1.15)$$

```
> select(t -> not has(t, I), {%});
```

$$\left\{-\frac{1}{2}\,2^{3/4},\ \frac{1}{2}\,2^{3/4}\right\} \quad (1.16)$$

```
> x1 := %[1]; x2 := %%[2];
```

$$x1 := -\frac{1}{2}\,2^{3/4}$$

$$x2 := \frac{1}{2}\,2^{3/4} \quad (1.17)$$

```
> allvalues(subs(M3, y));
```

$$\frac{1}{2}\,2^{3/4},\ \frac{1}{2}\,I\,2^{3/4},\ -\frac{1}{2}\,2^{3/4},\ -\frac{1}{2}\,I\,2^{3/4} \quad (1.18)$$

```
> select(t -> not has(t, I), {%});
```

$$\left\{-\frac{1}{2}\,2^{3/4},\ \frac{1}{2}\,2^{3/4}\right\} \quad (1.19)$$

```
> y1 := %[1]; y2 := %%[2];
```

$$y1 := -\frac{1}{2}\,2^{3/4}$$

$$y2 := \frac{1}{2}\,2^{3/4} \quad (1.20)$$

```
> allvalues(subs(M3, lambda));
```

$$\frac{1}{4}\,2^{3/4},\ \frac{1}{4}\,I\,2^{3/4},\ -\frac{1}{4}\,2^{3/4},\ -\frac{1}{4}\,I\,2^{3/4} \quad (1.21)$$

```
> select(t -> not has(t, I), {%});
```

$$\left\{-\frac{1}{4}\,2^{3/4},\ \frac{1}{4}\,2^{3/4}\right\} \quad (1.22)$$

```
> lambda1 := %[1]; lambda2 := %%[2]; # 2. Frage
```

$$\lambda 1 := -\frac{1}{4}\,2^{3/4}$$

$$\lambda 2 := \frac{1}{4}\,2^{3/4} \quad (1.23)$$

```
> it := 1:for x in [x1, x2] do;
>   for y in [y1, y2] do;
>     for lambda in [lambda1, lambda2] do;
>       print(it,x, y, lambda, GLF); it := it +1;
```

```
>     od;
>   od;
> od;
> x := 'x': y := 'y': lambda := 'lambda':
```

$$1, -\frac{1}{2}\,2^{3/4}, -\frac{1}{2}\,2^{3/4}, -\frac{1}{4}\,2^{3/4}, \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

$$2, -\frac{1}{2}\,2^{3/4}, -\frac{1}{2}\,2^{3/4}, \frac{1}{4}\,2^{3/4}, \begin{bmatrix} 2 \\ 2 \end{bmatrix}$$

$$3, -\frac{1}{2}\,2^{3/4}, \frac{1}{2}\,2^{3/4}, -\frac{1}{4}\,2^{3/4}, \begin{bmatrix} 0 \\ 2 \end{bmatrix}$$

$$4, -\frac{1}{2}\,2^{3/4}, \frac{1}{2}\,2^{3/4}, \frac{1}{4}\,2^{3/4}, \begin{bmatrix} 2 \\ 0 \end{bmatrix}$$

$$5, \frac{1}{2}\,2^{3/4}, -\frac{1}{2}\,2^{3/4}, -\frac{1}{4}\,2^{3/4}, \begin{bmatrix} 2 \\ 0 \end{bmatrix}$$

$$6, \frac{1}{2}\,2^{3/4}, -\frac{1}{2}\,2^{3/4}, \frac{1}{4}\,2^{3/4}, \begin{bmatrix} 0 \\ 2 \end{bmatrix}$$

$$7, \frac{1}{2}\,2^{3/4}, \frac{1}{2}\,2^{3/4}, -\frac{1}{4}\,2^{3/4}, \begin{bmatrix} 2 \\ 2 \end{bmatrix}$$

$$8, \frac{1}{2}\,2^{3/4}, \frac{1}{2}\,2^{3/4}, \frac{1}{4}\,2^{3/4}, \begin{bmatrix} 0 \\ 0 \end{bmatrix} \qquad \textbf{(1.24)}$$

```
> punkt1 := {x = x1, y = y1, lambda = lambda1};
```

$$punkt1 := \left\{\lambda = -\frac{1}{4}\,2^{3/4}, x = -\frac{1}{2}\,2^{3/4}, y = -\frac{1}{2}\,2^{3/4}\right\} \qquad \textbf{(1.25)}$$

```
> punkt2 := {x = x2, y = y2, lambda = lambda2};
```

$$punkt2 := \left\{\lambda = \frac{1}{4}\,2^{3/4}, x = \frac{1}{2}\,2^{3/4}, y = \frac{1}{2}\,2^{3/4}\right\} \qquad \textbf{(1.26)}$$

Test

```
> subs(punkt1, g);
```

$$0 \qquad \textbf{(1.27)}$$

```
> subs(punkt2, g);
```

$$0 \quad (1.28)$$

```
> H := Hessian[VectorCalculus](f-lambda*g,[x,y]);
```

$$H := \begin{bmatrix} -12\,\lambda\,x^2 & 0 \\ 0 & -12\,\lambda\,y^2 \end{bmatrix} \quad (1.29)$$

```
> subs(punkt1,H);
```

$$\begin{bmatrix} 3\,2^{1/4} & 0 \\ 0 & 3\,2^{1/4} \end{bmatrix} \quad (1.30)$$

Also haben wir hier ein Minimum (Achtung: Fuer das Minimum muss nur ein Teil der Hessematrix positiv definit sein
-> Optimierung

```
> subs(punkt2,H);
```

$$\begin{bmatrix} -3\,2^{1/4} & 0 \\ 0 & -3\,2^{1/4} \end{bmatrix} \quad (1.31)$$

Und hier ein Maximum

```
> wert := subs(punkt1, f);
```

$$wert := -2^{3/4} \quad (1.32)$$

```
> solve(f = wert, y);
```

$$-2^{3/4} - x \quad (1.33)$$

```
> pl_NFs := plot(%, x = -1.2 .. .1, thickness = 2):
> display({NB, pl_NFs}, scaling = constrained);
```



Extrema unter Nebenbedingungen in 3D

```
> f :=x^2+y^2+z^2;  # 3. Frage
```

$$f := x^2 + y^2 + z^2 \quad (2.1)$$

```
> g := x^2 + 2*y^2 - z^2 - 1;
```

$$g := x^2 + 2\,y^2 - z^2 - 1 \quad (2.2)$$

```
> with(plots):
> gp := implicitplot3d(g, x=-2..2, y=-2..2, z=-2..2, style =
  patchcontour, shading = zhue, scaling = constrained, axes =
  boxed, numpoints = 5000):
> gp;
```




```
> Nf3o2 := implicitplot3d(f=1.5, x = -1.5 .. 1.5, y = -1.5 ..
  1.5, z= -1.5..1.5, style = patchnogrid, color = black,
  transparency = .4,numpoints=6000):
> display({Nf3o2, gp},scaling = constrained);
```




```
> grad_f := Gradient(f, [x,y,z]);
```

$$grad\_f := \begin{bmatrix} 2x \\ 2y \\ 2z \end{bmatrix} \quad \textbf{(2.3)}$$

```
> grad_g := Gradient(g, [x, y, z]);
```

$$grad\_g := \begin{bmatrix} 2x \\ 4y \\ -2z \end{bmatrix} \quad \textbf{(2.4)}$$

```
> GLF := grad_f - lambda*grad_g;
```

$$GLF := \begin{bmatrix} -2\lambda x + 2x \\ -4\lambda y + 2y \\ 2\lambda z + 2z \end{bmatrix} \quad \textbf{(2.5)}$$

```
> Lsg := solve({GLF[1] = 0, GLF[2] = 0, GLF[3] = 0, g = 0}, {x,
```

```
y, z, lambda});
```

$$Lsg := \{\lambda = 1, x = 1, y = 0, z = 0\}, \{\lambda = 1, x = -1, y = 0, z = 0\}, \left\{\lambda = \frac{1}{2}, x = 0, y = RootOf(2\_Z^2 - 1), z = 0\right\}, \{\lambda = -1, x = 0, y = 0, z = RootOf(\_Z^2 + 1)\} \quad (2.6)$$

```
> allvalues(Lsg[4]);
```

$$\{\lambda = -1, x = 0, y = 0, z = I\}, \{\lambda = -1, x = 0, y = 0, z = -I\} \quad (2.7)$$

Loesungen sind imaginaer

```
> G1 := Lsg[1]; G2 := Lsg[2]; G3:=allvalues(Lsg[3]);
```

$$G1 := \{\lambda = 1, x = 1, y = 0, z = 0\}$$

$$G2 := \{\lambda = 1, x = -1, y = 0, z = 0\}$$

$$G3 := \left\{\lambda = \frac{1}{2}, x = 0, y = \frac{1}{2}\sqrt{2}, z = 0\right\}, \left\{\lambda = \frac{1}{2}, x = 0, y = -\frac{1}{2}\sqrt{2}, z = 0\right\} \quad (2.8)$$

```
> f1 := subs(G1, f); g1:=subs(G1,g);
```

$$f1 := 1$$

$$g1 := 0 \quad (2.9)$$

```
> f2 := subs(G2, f); g2:=subs(G2,g);
```

$$f2 := 1$$

$$g2 := 0 \quad (2.10)$$

```
> wert12 := f1;
```

$$wert12 := 1 \quad (2.11)$$

```
> ball1 := implicitplot3d(f-wert12, x = -1 .. 1, y = -1..1, z=-1..1, style = patchnogrid, color = black, transparency=0.2):
> display({gp, ball1}, scaling = constrained);
```




```
> f3:=subs(G3[1],f);
```

$$f3 := \frac{1}{2} \tag{2.12}$$

```
> f4:=subs(G3[2],f);
```

$$f4 := \frac{1}{2} \tag{2.13}$$

```
> ball2 := implicitplot3d(f-1/2, x = -1 .. 1, y = -1..1, z=-1..1,
  style = patchnogrid, color = black, transparency=0.2):
> display({gp, ball2}, scaling = constrained, orientation = [-30,
  35]);
```



Zeichenkettenverarbeitung und Plotverschoenerung

```
> restart:
> for j from 1 to 3 do;
>   a[j] := "Zeile Nummer "||j;
> od:
> a[3];
```

"Zeile Nummer 3" **(3.1)**

```
> z := "Zeile Nummer ";
```

z := "Zeile Nummer " **(3.2)**

```
> z||24;
```

z24 **(3.3)**

```
> ""||z||24;
```

"Zeile Nummer 24" **(3.4)**

```
> cat(z,24); # Verwenden sie statt || besser cat
```

"Zeile Nummer 24" **(3.5)**

```
> for j from 2 to 5 do;
>   Lsg := solve(x^j = 1, x):
>   print(Lsg[j-1]);
> od:
```

$$1$$

$$-\frac{1}{2} - \frac{1}{2} I\sqrt{3}$$

$$I$$

$$-\frac{1}{4}\sqrt{5} - \frac{1}{4} - \frac{1}{4} I\sqrt{2}\sqrt{5-\sqrt{5}} \quad (3.6)$$

```
> for j from 2 to 5 do;
>   Lsg := solve(x^j = 1, x):
>   print(cat(j," te Einheitswurzel:"), (Lsg[j-1]));
> od:
```

2 te Einheitswurzel:, 1

3 te Einheitswurzel:, $-\frac{1}{2} - \frac{1}{2} I\sqrt{3}$

4 te Einheitswurzel:, I

5 te Einheitswurzel:, $-\frac{1}{4}\sqrt{5} - \frac{1}{4} - \frac{1}{4} I\sqrt{2}\sqrt{5-\sqrt{5}}$ **(3.7)**

```
> f := x^2 - 1/2;
```

$$f := x^2 - \frac{1}{2} \quad (3.8)$$

```
> convert(f, string);
```

"x^2-1/2" **(3.9)**

```
> Beschreibung := "Graph von "||(convert(f, string));
```

Beschreibung := "Graph von x^2-1/2" **(3.10)**

```
> plot(f, x = -1 .. 1, title = Beschreibung, thickness = 2);
```




```
> Font := [TIMES, ROMAN, 24];
```

Font := [*TIMES*, *ROMAN*, 24] **(3.11)**

```
> plot(f, x = -1 .. 1, title = Beschreibung, font = Font,
  thickness = 2);
```




```
> Beschreibung2 := cat( Beschreibung," und ihrer Ableitung");
```

Beschreibung2 := "Graph von x^2-1/2 und ihrer Ableitung" **(3.12)**

```
> Optionen := title = Beschreibung2, font = Font, titlefont =
  Font, labels = [x,y], thickness = 2;
```

Optionen := *title* = "Graph von x^2-1/2 und ihrer Ableitung", *font* = [*TIMES*, *ROMAN*, 24], *titlefont* = [*TIMES*, *ROMAN*, 24], *labels* = [*x*, *y*], *thickness* = 2 **(3.13)**

```
> pl1:= plot([f, diff(f, x)], x = -1 .. 1, Optionen):
> pl1;
```




```
> with(plots):
> t1 := textplot([-.8, .2, "f"], align = {ABOVE, RIGHT}, font =
  Font):
> display({pl1, t1});
```




```
> position := -.6, subs(x = -.6, diff(f,x));
```

$position := -0.6, -1.2$ **(3.14)**

```
> t2 := textplot([position, "f'"], align = {BELOW, RIGHT}, font =
  Font):
> display({pl1, t1, t2}, font = Font);
```




```
> t3 := textplot3d([1, 0, 1, "Text schwebt im Raum"], font =
  Font, color = green):
> x := r*cos(theta);
```

$$x := r\cos(\theta) \tag{3.15}$$

```
> y := r*sin(theta);
```

$$y := r\sin(\theta) \tag{3.16}$$

```
> f := sqrt(1 - r^2) * (1 - sin(12*theta)/10);
```

$$f := \sqrt{-r^2+1}\left(1-\frac{1}{10}\sin(12\,\theta)\right) \tag{3.17}$$

```
> pl2 := plot3d([x, y, f], r = 0 .. 1, theta = 0 .. 2*Pi,
  lightmodel = light4, shading = zgrayscale, grid = [10, 200],
  style = patchcontour):
> display(pl2, t3);
```




```
> dateiname := "test":
> plotsetup(gif, plotoutput = dateiname);
> display(pl2, t3);
> plotsetup(window);
> display(pl2, t3);
```